The Triangular Squares

An elaboration of a brief email exchange with noted mathematician Professor Mike Hirschhorn, concerning the sequence 0, 1, 36, 1225, 41616, 1413721, 48024900, 1631432881, 55420693056, 1882672131025, 63955431761796, ..., the numbers which are simultaneously triangular and square.

Dear Bill,
If the p_k/q_k are the convergents to sqrt{2}, then (p_kq_k)^2 is a triangular number, and these are the only numbers that are both squares and triangles. They satisfy the four-term recurrence x_k=35x_{k-1}-35x_{k-2}+x_{k-3}.

Hi Mike, alternatively the three term inhomogeneous:

$$x(n) = 2 + 34\,x(n-1) - x(n-2)$$

Best regards, Mike

As starkly as possible,

$$\left(\binom{\frac{(\sqrt{2}+1)^n+(\sqrt{2}-1)^n}{2}}{2}\right) = \left(\frac{(\sqrt{2}+1)^{2n}-(\sqrt{2}-1)^{2n}}{4\sqrt{2}}\right)^2$$

```
Simplify[ReleaseHold[%]]
```

True

But my point was, you needn't know all this. You can just take a 4◊4 determinant of any six consecutive values, and extend the sequence *quodlibet*. And it's easy to get started with the conveniently bunched values on both sides of $n = 0$, choosing the first few squares that happen to be triangular.

But *not* by choosing the first few triangulars that happen to be square!

Even simpler: Fred Lunnon points out you only need four consecutive values and 3◊3 determinants to extend the sequence of signed square roots $\sqrt{x(n)}$ = ..., - 6, - 1, 0, 1, 6, 35, 204, 1189, 6930, ...

Comparably superlative regards,
Bill

First, let's test Mike's claim about the convergents to $\sqrt{2}$.

In[80]:=
```
Convergents[√2, 9]
```

Out[80]= $\left\{1, \frac{3}{2}, \frac{7}{5}, \frac{17}{12}, \frac{41}{29}, \frac{99}{70}, \frac{239}{169}, \frac{577}{408}, \frac{1393}{985}\right\}$

These are the approximations to $\sqrt{2}$ which are best for their size:

In[82]:=
```
%80^2
```

Out[82]= $\left\{1, \frac{9}{4}, \frac{49}{25}, \frac{289}{144}, \frac{1681}{841}, \frac{9801}{4900}, \frac{57121}{28561}, \frac{332929}{166464}, \frac{1940449}{970225}\right\}$

In[83]:=
```
2 - %
```

Out[83]= $\left\{1, -\frac{1}{4}, \frac{1}{25}, -\frac{1}{144}, \frac{1}{841}, -\frac{1}{4900}, \frac{1}{28561}, -\frac{1}{166464}, \frac{1}{970225}\right\}$

He claims that the numerators times the denominators of these approximations to 2 are exactly the triangular squares:

In[84]:=
```
Numerator[%%] * Denominator[%%]
```

Out[84]= {1, 36, 1225, 41616, 1413721, 48024900, 1631432881, 55420693056, 1882672131025}

Familiar? See http://oeis.org/A001110 for lots of formulas and references. It will turn out that the square roots of these,

In[85]:=
```
√%
```

Out[85]= {1, 6, 35, 204, 1189, 6930, 40391, 235416, 1372105}

i.e., those numbers which become triangular when squared, obey a simpler recurrence formula. (See http://oeis.org/A001109 .)

Note my "starkly as possible" answer. It's pretty stiff algebra to confirm that the two sides are algebraically equivalent. The equation is between the binomial of something and the square of something else. It is worth your while to see why these somethings are always integers.

Mathematica directly (and rather impressively) finds the general formula for which triangular number equals which square by using Pell's equation:

In[77]:=
```
Reduce[j^2 == k * (k + 1) / 2, {k, j}, Integers] // TraditionalForm
```

Out[77]//TraditionalForm=

$$\left(c_1 \in \mathbb{Z} \land c_1 \geq 0 \land k = \frac{1}{2}\left(\frac{1}{2}\left((3-2\sqrt{2})^{c_1}+(3+2\sqrt{2})^{c_1}\right)-1\right) \land j = \frac{(3-2\sqrt{2})^{c_1}-(3+2\sqrt{2})^{c_1}}{4\sqrt{2}}\right) \lor$$
$$\left(c_1 \in \mathbb{Z} \land c_1 \geq 0 \land k = \frac{1}{2}\left(\frac{1}{2}\left(-(3-2\sqrt{2})^{c_1}-(3+2\sqrt{2})^{c_1}\right)-1\right) \land j = \frac{(3-2\sqrt{2})^{c_1}-(3+2\sqrt{2})^{c_1}}{4\sqrt{2}}\right) \lor$$
$$\left(c_1 \in \mathbb{Z} \land c_1 \geq 0 \land k = \frac{1}{2}\left(\frac{1}{2}\left((3-2\sqrt{2})^{c_1}+(3+2\sqrt{2})^{c_1}\right)-1\right) \land j = -\frac{(3-2\sqrt{2})^{c_1}-(3+2\sqrt{2})^{c_1}}{4\sqrt{2}}\right) \lor$$
$$\left(c_1 \in \mathbb{Z} \land c_1 \geq 0 \land k = \frac{1}{2}\left(\frac{1}{2}\left(-(3-2\sqrt{2})^{c_1}-(3+2\sqrt{2})^{c_1}\right)-1\right) \land j = -\frac{(3-2\sqrt{2})^{c_1}-(3+2\sqrt{2})^{c_1}}{4\sqrt{2}}\right)$$

c_1 is an arbitrary integer constant ..., -1, 0, 1, 2, ... which we shall rename n.

In[86]:=
```
FullSimplify[%77] /. a_ == b_ :> Solve[a == b, {j, k}] /. C[1] → n
```

Out[86]=
$$n \in \text{Integers} \;\&\&\; \left(\left\{\left\{j \to -\frac{(3-2\sqrt{2})^n-(3+2\sqrt{2})^n}{4\sqrt{2}}\right\}\right\} \,||\, \left\{\left\{j \to \frac{(3-2\sqrt{2})^n-(3+2\sqrt{2})^n}{4\sqrt{2}}\right\}\right\}\right) \;\&\&$$
$$\left(\left\{\left\{k \to \frac{1}{4}\left(-2-(3-2\sqrt{2})^n-(3+2\sqrt{2})^n\right)\right\}\right\} \,||\, \left\{\left\{k \to \frac{1}{4}\left(-2+(3-2\sqrt{2})^n+(3+2\sqrt{2})^n\right)\right\}\right\}\right) \;\&\&\; n \geq 0$$

If FullSimplify were really smart, it would discard the &&n≥0 and one of the two j solutions and one of the two k solutions, which are merely swapped if n->-n.

The purpose of this note is not Pell's equation, but rather to give a more elementary solution, once we propose that the triangular squares obey a linear recurrence. The recurrence I gave needs only two consecutive values, say 0 and 1, to produce all the rest:

In[14]:=
```
Nest[Append[#, 2 + 34 * #[[-1]] - #[[-2]]] &, {0, 1}, 9]
```

Out[14]= {0, 1, 36, 1225, 41616, 1413721, 48024900, 1631432881, 55420693056, 1882672131025, 63955431761796}

But, as I mentioned, their square roots

In[16]:=
```
√%14
```

Out[16]= {0, 1, 6, 35, 204, 1189, 6930, 40391, 235416, 1372105, 7997214}

obey a simpler recurrence. If we knew what it was, we'd only need 0 and 1 to get the rest. Or better yet, *solve* the recurrence, getting them all at once. We can do all this knowing only the fragment ..., 0, 1, 6, ... because we can extend it to negative: ..., -6, -1, 0, 1, 6, ..., and, for *any* second order recurrence (of this type), four consecutive values determine the next, because the determinant formed from five consecutive terms vanishes:

In[17]:=
```
Solve[0 == Det[{{-6, -1, 0}, {-1, 0, 1}, {0, 1, x}}]]
```

Out[17]= {{x → 6}}

In[18]:=
```
Solve[0 == Det[{{-1, 0, 1}, {0, 1, 6}, {1, 6, x}}]]
```

Out[18]= {{x → 35}}

Given a fairly short burst of values, *Mathematica* (correctly) guesses the general formula.

In[19]:=
```
FindSequenceFunction[{1, 6, 35, 204, 1189, 6930, 40391}, n]
```

Out[19]= $-\frac{(4+3\sqrt{2})\left((3-2\sqrt{2})^n-(3+2\sqrt{2})^n\right)}{8(3+2\sqrt{2})}$

In[20]:=
```
FullSimplify[%, n ∈ Integers]
```

Out[20]= $\frac{-(3-2\sqrt{2})^n+(3+2\sqrt{2})^n}{4\sqrt{2}}$

Instead of the "stiff" algebra of directly verifying the equality of the triangular and square expressions, it is much easier to verify that they individually satisfy the aforementioned second order recurrence:

In[26]:=
```
Simplify[
 x(n) ‡ 2 + 34 x(n - 1) - x(n - 2) /. x[n_] -> Binomial[((√2 + 1)^n + (√2 - 1)^n)/2, 2] //
  ReleaseHold]
```

Out[26]= True

In[27]:=
```
Simplify[
 x(n) ‡ 2 + 34 x(n - 1) - x(n - 2) /.
  x[n_] -> (((√2 + 1)^(2 n) - (√2 - 1)^(2 n))/(4 √2))^2 // ReleaseHold]
```

Out[27]= True

Then we need only check that both expressions start out 0 and 1 for n = 0 and 1:

In[22]:=
```
Table[Binomial[((√2 + 1)^n + (√2 - 1)^n)/2, 2] -> (((√2 + 1)^(2 n) - (√2 - 1)^(2 n))/(4 √2))^2 //ReleaseHold, {n, 0, 2}]
```

Out[22]= $\left\{0 \to 0,\; 1 \to \frac{1}{32}\left(-(-1+\sqrt{2})^2+(1+\sqrt{2})^2\right)^2,\; \frac{1}{8}\left((-1+\sqrt{2})^2+(1+\sqrt{2})^2\right)^2\left(-1+\frac{1}{4}\left((-1+\sqrt{2})^2+(1+\sqrt{2})^2\right)^2\right) \to \frac{1}{32}\left(-(-1+\sqrt{2})^4+(1+\sqrt{2})^4\right)^2\right\}$

In[23]:=
```
Simplify[%]
```

Out[23]= {0 → 0, 1 → 1, 36 → 36}

The recurrence then constrains them to agree forever.

But how does FindSequence function guess recurrence formulæ? (It has to be guessing, for nothing constrains us from throwing in crazy values later in the data.) Suppose we group the triplets of consecutive terms of 0, 1, 6, 35, ... into 2◊2 matrices:

In[36]:=
```
Table[
 MatrixForm[HankelMatrix[Take[%16, {k, k + 1}],
   Take[%16, {k + 1, k + 2}]]], {k, 3}]
```

Out[36]= $\left\{\begin{pmatrix}0 & 1\\1 & 6\end{pmatrix}, \begin{pmatrix}1 & 6\\6 & 35\end{pmatrix}, \begin{pmatrix}6 & 35\\35 & 204\end{pmatrix}\right\}$

The recurrence can be mechanized by a magic matrix M which advances each sequence matrix to the next. E.g., for the first step, at least,

In[37]:=
```
M.%[[1]] == %[[2]]
```

Out[37]= $M.\begin{pmatrix}0 & 1\\1 & 6\end{pmatrix} == \begin{pmatrix}1 & 6\\6 & 35\end{pmatrix}$

What M does this? Right-multiply both sides by the inverse of $\begin{pmatrix}0 & 1\\1 & 6\end{pmatrix}$:

In[38]:=
```
MapAt[Inverse, %[[1, 2]], 1]
```

Out[38]//MatrixForm=
$\begin{pmatrix}-6 & 1\\1 & 0\end{pmatrix}$

In[88]:=
```
%37 /. MatrixForm[m_] :> MatrixForm[m.%38]
```

Out[88]= $M.\begin{pmatrix}1 & 0\\0 & 1\end{pmatrix} == \begin{pmatrix}0 & 1\\-1 & 6\end{pmatrix}$

I.e., the advancer matrix claims to be

In[90]:=
```
%88 /. Dot → (# &)
```

Out[90]= $M == \begin{pmatrix}0 & 1\\-1 & 6\end{pmatrix}$

Does it repeadly advance $\begin{pmatrix}0 & 1\\1 & 6\end{pmatrix}$?

In[95]:=
```
MatrixForm /@ NestList[{{0, 1}, {-1, 6}}.# &, {{0}, {1}}, 6]
```

Out[95]= $\left\{\begin{pmatrix}0\\1\end{pmatrix}, \begin{pmatrix}1\\6\end{pmatrix}, \begin{pmatrix}6\\35\end{pmatrix}, \begin{pmatrix}35\\204\end{pmatrix}, \begin{pmatrix}204\\1189\end{pmatrix}, \begin{pmatrix}1189\\6930\end{pmatrix}, \begin{pmatrix}6930\\40391\end{pmatrix}\right\}$

Thus if f(n=0,1,2,...) = 0, 1, 6, ..., advancement looks like

Out[56]= $\begin{pmatrix}f[1+n] & f[2+n]\\f[2+n] & f[3+n]\end{pmatrix} == \begin{pmatrix}0 & 1\\-1 & 6\end{pmatrix}.\begin{pmatrix}f[n] & f[1+n]\\f[1+n] & f[2+n]\end{pmatrix}$

In[62]:=
```
MapAt[Thread[#, MatrixForm] &, %56, 2]
```

Out[62]= $\begin{pmatrix}f[1+n] & f[2+n]\\f[2+n] & f[3+n]\end{pmatrix} == \begin{pmatrix}f[1+n] & f[2+n]\\-f[n]+6f[1+n] & -f[1+n]+6f[2+n]\end{pmatrix}$

Equating lower left elements gives the recurrence

In[68]:=
```
#[[1, 2, 1]] & /@ %62
```

Out[68]= f[2 + n] == - f[n] + 6 f[1 + n]

Solving such (linear, constant coefficient) recurrences is routine:

In[69]:=
```
RSolve[{%, f[0] == 0, f[1] == 1}, f[n], n]
```

Out[69]= $\left\{\left\{f[n] \to -\frac{(3-2\sqrt{2})^n-(3+2\sqrt{2})^n}{4\sqrt{2}}\right\}\right\}$

Hint: "Factor" the denominator of the generating function Sum(f(n) x^n) down to its roots, expand in partial fractions, and then into geometric series.

P.S., the numbers whose trianglings are square:

In[96]:=
```
-InverseFunction[# * (# + 1) / 2 &] /@
  {0, 1, 36, 1225, 41616, 1413721, 48024900, 1631432881,
   55420693056, 1882672131025, 63955431761796}
```

InverseFunction::ifun :
Inverse functions are being used. Values may be lost for multivalued inverses. ≫

Out[96]= {1, 2, 9, 50, 289, 1682, 9801, 57122, 332929, 1940450, 11309769}

(http://oeis.org/A055997)